Mplus syntax for the estimation of the internal consistency of essentially undimensional measures according to two compatible measurement models as described in
Viladrich, C., Angulo-Brunet, A., & Doval, E. (2017). A journey around alpha and omega to estimate internal consistency reliability. Annals of Psychology, 33(3), 755-782. doi: 10.6018/analesps.33.3.268401
See http://ddd.uab.cat/record/173917 for the dataset headed by the variable names and Table 1 and Table 2 in the paper for selected output obtained running the first measurement model with R.
See http://ddd.uab.cat/record/205870 for the dataset without variable names, as required by Mplus. Output using Mplus could differ from Table 1 and Table 2 due to differences in computational algorithms.

Table of contents


```
!Starting Mplus
!Defining the working directory
!Save the input file in the same folder as the data file

!Installing and checking packages needed to perform the analyses
!All analyses use the Mplus Base Program
!Copy-paste each syntax from TITLE to OUTPUT to separate Mplus input files
!(extension .inp)


!Case 3: Essentially unidimensional measures
!Phase 1
TITLE:
  Response percentiles, other univariate statistics, Pearson correlations
DATA:
  FILE IS Case3_noNames.txt;
VARIABLE:
  NAMES = y1-y6;
  USEVARIABLES = y1-y6;
ANALYSIS: TYPE = BASIC;
OUTPUT: sampstat


!Measures with correlated errors
!Phase 2
TITLE:
  Specification, estimation and fit of measures with correlated errors
DATA:
  FILE IS Case3_noNames.txt;
VARIABLE:
  NAMES =  y1-y6 ;
  USEVARIABLES =  y1-y6 ;
MODEL:
  F1 by y1-y6*;
  F1@1;
  y4 with y5 y6;
  y5 with y6;
OUTPUT: stdyx tech1

!Phase 3
TITLE:
  Point and interval estimation of coefficient omega
DATA:
  FILE IS Case3_noNames.txt;
VARIABLE:
  NAMES = y1-y6;
```

```
  USEVARIABLES = y1-y6;
  ANALYSIS: BOOTSTRAP = 500;
MODEL:
  F1 by y1-y6* (la1-la6);
  F1@1;
  y4 with y5 (s45);
  y4 with y6 (s46);
  y5 with y6 (s56);
   y1-y6 (e1-e6);
MODEL CONSTRAINT:
!equation 5 en Viladrich, Angulo-Brunet y Doval (2017)
NEW(omega num, den);
num =  (la1+la2+la3+la4+la5+la6)^2;
den = num + e1+e2+e3+e4+e5+e6+ 2*(s45+s46+s56);
omega = num/den;
OUTPUT: CINTERVAL(BCBOOTSTRAP);


!Confirmatory bifactor measurement model
!Phase 2
TITLE:
  Specification, estimation and fit of the confirmatory bifactor model
DATA:
  FILE IS Case3_noNames.txt;
VARIABLE:
  NAMES =  y1-y6 ;
  USEVARIABLES =  y1-y6 ;
MODEL:
  Fg by y1* y2-y6;
  Fg@1;
  Fs by y4* y5 y6;
  Fs@1;
  Fg with Fs@0;
OUTPUT: stdyx tech1

!Phase 3
TITLE:
  Point and interval estimation of coefficients omega hierarchical and
  omega total
DATA:
  FILE IS Case3_noNames.txt;
VARIABLE:
  NAMES =  y1-y6 ;
  USEVARIABLES =  y1-y6 ;
ANALYSIS: BOOTSTRAP = 500;
MODEL:
  Fg by y1*(a1)
  y2-y6 (a2-a6);
  Fg@1;
  Fs by y4* (b4)
  y5 y6 (b5 b6);
  Fs@1;
  Fg with Fs@0;
  y1-y6 (e1-e6);
  MODEL CONSTRAINT:
  NEW(spec gen6 resid6 omegaHg omegaTg);
  spec=(b4+b5+b6)^2;
  gen6=(a1+a2+a3+a4+a5+a6)^2;
  resid6=e1+e2+e3+e4+e5+e6;
  omegaHg=gen6/(gen6+spec+resid6); !eq 8 Viladrich,Angulo-Brunet&Doval(2017)
  omegaTg=(gen6+spec)/(gen6+spec+resid6); !eq 9 Viladrich et al (2017)
OUTPUT: CINTERVAL(BCBOOTSTRAP);
```